Modify()

Objectifs de Modify

La Fonction Modify modifie un objet DataWindow en appliquant les spécifications définies sous forme d’une liste d’instructions transmise en argument de la fonction qui modifie la définition de l’objet DataWindow.

On peut changer le comportement, l’apparence, et les informations de la base de l’objet DataWindow en modifiant les valeurs de ses attributs.

On peut également ajouter des objets dans la DataWindow en donnant les spécification du nouvel objet et supprimer des objets de la DataWindow.

Cette fonction peut être appliquée à un contrôle DataWindow ou à une fenêtre enfant DataWindow.

Syntaxe:

DataWindowname.Modify(ModString)

	Parametre
	Description

	datawindowname
	Le Nom du contrôle DataWindow ou de la fenêtre enfant DataWindow que l’on modifie.

	modstring 
	Une chaîne contenant les spécifications des modifications. 


Valeur de Retour

String.

Renvoie une chaîne vide("") si succès.

Et un message d’erreur si une erreur survient.

Le message d’erreur est sous la forme "Line n Column n incorrect syntax".

Les colonnes sont comptées à partir du début de la partie compilable du texte de la chaîne modstring.

Utilisation du modify

Modify permet de définir des mêmes caractéristiques dans un script que l’on fait généralement à travers le DataWindow painter.

Les utilisations du Modify sont typiquement les suivantes :

Changement des couleurs, de la définition du texte, et autres définitions de l’apparences de l’objet

Changement du statut de mise à jour(update) des différentes tables de la DataWindow, afin de pouvoir modifier plus d’une table.

Changement de la clause WHERE de la DataWindow de l’instruction SQL SELECT.

Changement du mode de requête ou possibilité de permettre à l’utilisateur de définir des critères de recherche afin d’obtenir dynamiquement les données qu’il désire.

Changement du statut de l’option Retrieve Only As Needed.

Changement de la source de donnée de l’objet DataWindow.

Contrôle de l’aperçu avant impression.

Suppression et ajout d’objets,(ligne ou bitmaps dans l’objet DataWindow).

On peut utiliser trois types d’instructions dans modstring pour modifier un objet DataWindow.

	Type d’instructions
	Utiliser pour

	CREATE objet(spécifications)
	Ajoute des objets à l’objet DataWindow(Tel que du texte, champs calculés, et bitmaps) .

Les spécifications sont une liste d’attributs et valeurs utilisant le format décrit ci-dessous.

Pour créer un objet il faudra indiquer un nombre suffisant d’information permettant de le définir.

	DESTROY [COLUMN] objet
	Supprime un objet de l’objet DataWindow. Lorsque l’objet est un nom de colonne, spécifier le mot clé COLUMN pour supprimer à la fois la colonne et la zone de données de la colonne dans le buffer.

	objectname.attribute = value
	Change la valeur de l’attribut par valeur. Les attributs contrôlent la position, la couleur, la taille, la police(font), et autres définitions de l’objet designé par objectname.

Lorsque l’objectname est la DataWindow, on peut également définir les attributs de l’accès à la base.

Dans le paragraphe ci dessous nous listons les objets d’une DataWindow, leurs attributs et leurs valeurs.

L’attribut, value peut être

- Une constante

- Une constante entre quotes

- Une expression qui consiste à une valeur par défaut suivie d’une expression valide de DataWindow qui renvoie le type de donnée approprié de l’attribut.


Attributs d’Objet d’une DataWindow

Les attributs d’objets d’une DataWindow contrôlent l’apparence et le comportement d’une DataWindow.

Il y a plusieurs façon de modifier ces valeurs durant l’exécution du programme.

Il y a plusieurs fonctions qui permettent d’obtenir et de modifier la valeur des attributs.(Modify, Describe , SyntaxFromSQL) .

Mots clés

Group

Style

Title

Objets

Bitmap

Column

Computed field

DataWindow

Graph

Line

Oval, Rectangle, and RoundRectangle

Report

TableBlob

Text

Nom D’objet

Le DataWindow painter attribue automatiquement des noms aux colonnes et aux labels de colonne.

Les autres objets ajoutés par l’utilisateur sont nommés par une chaîne cryptée formée de nombres, sauf si l’utilisateur attribue lui même un nom.

Describe peut faire connaître le nom d’un objet. Mais ce nom n’est pas visible à travers le painter.

Pour conserver les facilités de modification des attributs des objets donner les noms à aux objets.

Expression pour la modification

Lorsque l’on spécifie une expression pour un attribut d’une DataWindow, l’expression a le format suivant :

ValeurParDefaut~tExpressionDuPainterDeDataWindow

ValeurParDefaut est une valeur qui peut être convertie dans le type approprié pour l’attribut.

Il est suivi d’une tabulation(~t).

ExpressionDuPainterDeDataWindow est une expression qui peut utiliser toute fonction du painter DataWindow.

L’expression doit être évaluée de telle sorte qu’elle corresponde au type de donnée de l’attribut.

Lorsqu’on attribue des valeurs aux attributs de colonne, l’expression est évaluée pour chaque ligne dans la DataWindow.

Ce qui permet de pouvoir changer l’apparence de cellule des données en fonction des valeurs des données.

Une expression typique utilisant la fonction If :

'16777215 ~t If(emp_status=~~'A~~',255,16777215)'

Pour utiliser cette expression dans la chaîne modstring, spécifier l’instruction suivante

modstring = "emp_id.Color='16777215 ~tIf(emp_status=~~'A~~',255,16777215)'"

Tous les attributs n’acceptent pas des expressions.

Quotes et tildes

Parce que les arguments de Modify sont présentés sous forme de chaîne qui peut inclure une autre chaîne, on est obligé d’utiliser une syntaxe spéciale pour spécifier une marque de quotation(les guillemets ou les quotes).

Pour spécifier qu’une marque de quotation est utilisée dans la chaîne, on peut soit spécifier l’autre style de quotes(simple si on utilisait les doubles, doubles si on utilisait des simples) ou précéder la marque de quotation par un tilde(~).

Pour une spécification de chaîne incorporée elle même ~", on doit spécifier ~~(qui spécifie un tilde) suivi par ~"(qui spécifie une quote).

Par exemple, une autre manière de définir l’instruction précédente est décrite ci dessous.

modstring = "emp_id.Color=~"16777215 ~tIf(emp_status=~~~"A~~~",255,16777215)~""

Construction de modstring avec des variables

Pour utiliser des variables de données dans un modstring, on peut définir la chaîne en utilisant des variables du programme.

Lorsque l’on concatène des parties de modstring, faire la vérification que les quotes ont été introduites aux endroits nécessaires.

Par exemple:

Le code suivant construit un modstring similaire aux chaînes précédemment évoquées, mais la valeur par défaut de la couleur et les deux valeurs de couleur de la fonction if sont calculés dans le script.

Noter que les quotes simples autour de l’expression entourent la chaîne.

red_amount = Integer(sle_1.Text)

modstring = "emp_id.Color='" + &


String(RGB(red_amount, 255, 255)) + &


"~tIf(emp_status=~~'A~~'," + &


String(RGB(255, 0, 0)) + &


"," + &


String(RGB(red_amount, 255, 255)) + &


")'"

L’exemple suivant n’utilise pas la fonction If.

On n’a pas besoin de quotes autour de la valeur, si l’on

n’utilise pas d’expression.

modstring = "emp_id.Color=" + String(RGB(red_amount, 255, 255))

On peut modifier plusieurs attributs avec un appel unique a la fonction Modify en incluant chaque définition de valeur d’attribut dans la chaîne modstring, separée par des espaces.

rtn = dw_1.Modify("emp_id.Font.Italic=0  &





oval_1.Background.Mode=0  &





oval_1.Background.Color=255")

Il est plus facile de comprendre et de debugger un programme pour lequel un appel à modify modifie un et un seul attribut.

Debug

Si l’on construit modstring à travers une variable, qui sera argument de la fonction modify, on peut visualiser le contenu de la variable en mode debug.

Lorsque le message d’erreur précise une colonne, on peut compter les caractères en observant la chaîne modstring.

Modification de la clause where

Utiliser la fonction modify pour modifier la clause where d’une requête SQL en lieu et place de la fonction SetSQLSelect.

La fonction Modify vérifie la syntaxe une fois seulement et ne change pas le statut update de l’objet DataWindow.

La fonction SetSQLSelect modifie la syntaxe deux fois(quand la syntaxe est modifiée(lorsque les valeurs sont demandées Exécution du Retrieve) et affecte le statut update de l’objet DataWindow.

PowerBuilder fournit plusieurs fonctions qui permettent de modifier une DataWindow.

Avant d’utiliser la fonction DataWindow , érifier s’il n’existe pas une fonction des objets DataWindow et/ou du contrôle DataWindow capable d’effectuer la modification.

La fonction Modify permet de modifier les attributs d’un objet DataWindow.

On peut modifier la valeur des attributs d’un contrôle DataWindow qui contiennent l’objet en utilisant la notation standard.

Par exemple pour définir une bordure autour du contrôle spécifier:

dw_1.Border = TRUE

ILLUSTRATION DE modify

Ces exemples illustrent l’utilisation typique des caractéristiques de la DataWindow.

Changement des couleurs

L’effet du changement des attributs de couleurs dépend des objets manipulés.

Pour modifier la couleur de fond d’une fenêtre de l’objet DataWindow, utiliser la syntaxe suivante :

dwcontrôlename.Modify("DataWindow.Color='long'")

Pour définir la couleur du texte d’une colonne ou d’un objet texte, utiliser une syntaxe similaire.

dwcontrôlename.Modify("objectname.Color='long'")

Pour définir la couleur d’arrière plan d’une colonne ou d’un autre objet, utiliser la syntaxe suivante pour modifier le mode et la couleur. Assurer vous que le mode est opaque.

dwcontrôlename.Modify( &


"objectname.Background.Mode='<0 - Opaque, 1 - Transparent>'")

dwcontrôlename.Modify("objectname.Background.Color='long'")

Les exemples suivants utilisent la syntaxe vue précédemment pour définir des couleurs de differentes parties de l’objet DataWindow.

Cette instruction change la couleur d’arrière plan de la DataWindow à rouge.

dw_cust.Modify("DataWindow.Color = 255")

Cette instruction permet d’afficher le texte des valeurs dans la colonne salaire en rouge d’une DataWindow dw_cust si elle dépasse la valeur de 90000.

dw_cust.Modify( &


"salary.Color='0~tIf(salary>90000,255,65280)'")

Cette instruction inclut une instruction If dans une autre instruction If pour donner la possibilité de choix entre 3 colonnes. Les instructions permettent d’afficher les valeurs de la colonne département de la DataWindow dw_cust en vert si l’ID est 200, en rouge si l’ID est 100, et en noir sinon.

dw_cust.Modify("dept_id.Color='0~t "&


+ "If(dept_id=200,65380,If(dept_id=100,0,255))'")

L’instruction suivante utilise une expression complexe avec des if imbriqués pour définir la couleur d’arrière plan de la colonne salaire en fonction de ses valeurs.

Chaque partie de chaîne à fabriquer est visualisée sur une ligne séparée.

L’exemple définit aussi le mode d’arrière plan à opaque, de telle sorte que les définitions de couleurs sont visibles. Les exemples inclus l’analyse de l’erreur générée par la fonction Modify.

string mod_string, err

long color1, color2, color3, default_color

err = dw_emp.Modify("salary.Background.Mode=0")

IF err <> "" THEN


MessageBox("Status",  &



"Change to Background Mode Failed " + err)


RETURN

END IF

/* Pseudo code pour mod_string:

Si salaire inférieur à 10000 définir la couleur d’arriere plan à rouge

Si salaire supérieur à 10000 mais inférieur à 20000, définir la couleur d’arrière plan à bleu

Si salaire supérieur à 10000 mais inférieur à 20000, définir la couleur d’arrière plan à vert

Sinon définir la couleur d’arrière plan à blanc qui est aussi la valeur par défaut.

*/

color1 = 255  



//rouge

color2 = 16711680 


//bleu

color3 = 65280 



//vert

default_color = 16777215

//blanc

mod_string = &

"salary.Background.Color = '"  &


+ String(default_color) &


+ "~tIf(salary < 10000,"  &


+ String(color1) &


+ ",If(salary < 20000,"  &


+ String(color2) &


+ ",If(salary < 30000,"  &


+ String(color3) &


+ ","  &


+ String(default_color) &


+ ")))'"

err = dw_emp.Modify(mod_string)

IF err <> "" THEN


MessageBox("Status", &



"Change to Background Color Failed " + err)


RETURN

END IF

L’exemple suivant définit la couleur du texte d’une colonne radio button à la valeur de color1 qui est rouge, si la valeur de la colonne est Y. Sinon la couleur du texte sera noire

Comme précédemment chaque partie de la chaîne concatenée est indiquée sur une ligne différente:

integer color1, default_color

string mod_string, err

color1 = 255            //rouge

default_color = 0       //noire

mod_string = "yes_or_no.Color ='"  &

   + String(default_color) &

   + "~tif(yes_or_no=~~'Y~~',"  &

   + String(color1) &

   + ","  &

   + String(default_color) &

   + ")'"

err = dw_emp.Modify(mod_string)

IF err <> "" THEN

   MessageBox("Status",  &

   "Modify to Text Color "  &

   + "of yes_or_no Failed " + err)

   RETURN

END IF

Changement du texte affiché

Pour définir le texte d’un objet texte, les deux exemples qui suivent, utilisent la syntaxe suivante.

dwcontrôlename.Modify("textobjectname.Text='string'")

Cette instruction change le texte dans l’objet texte

Dept_t de la DataWindow dw_cust à Dept.


dw_cust.Modify("Dept_t.Text='Dept'")

Cette instruction change le texte dans l’objet texte Dept_t dans la DataWindow dw_cust à Marketing si le département ID est supérieur à 201 sinon définir le texte à la valeur Finance :

dw_cust.Modify("dept_t.Text='none~t "  &


+ "If(dept_id > 201,~'Marketing~',~'Finance~')'")

Mise à jour de plus d’une table.

Une utilisation importante de la fonction modify est de rendre possible la modification de plus d’une table de l’objet DataWindow.

Le script suivant met à jour la table qui était spécifiée comme pouvant être mise à jour dans le DataWindow painter.

Ensuite dans le script on utilise la fonction modify pour rendre possible la mise à jour des autres tables de la jointure, et pour spécifier la colonne clé et les colonnes qui peuvent être mise à jour.

Cette technique élimine la nécessité de créer plusieurs objets DataWindow ou d’utiliser des instructions embedded sql pour faire la mise à jour de plus d’une table.

Dans l’exemple, l’objet DataWindow joint deux tables département et employée. Le département est d’abord mis à jour sans modifier le statut update.

Le statut de la table employée est modifié pour rendre la mise à jour possible, et la table est ensuite mise à jour.

Si toutes les commandes se déroulent avec succès la commande Update met à jour le statut de mise à jour et lance le commit pour valider les modifications effectuées.

Pour donner la possibilité d’exécuter le script plusieurs fois dans une session, on doit ajouter au code l’obligation de faire en sorte que la table département puisse de nouveau être mise à jour.

integer rc

string err

/* L’instruction select de la datawindows est la suivante:

SELECT department.dept_id, department.dept_name, employee.emp_id,employee.emp_fname, employee.emp_lname FROM department, employee ;

*/

// Mettre à jour département comme spécifié dans le DataWindow painter

rc = dw_1.Update(TRUE, FALSE)

IF rc = 1 THEN

   //Mettre à No le statut de mise à jour des colonnes de department.

   dw_1.Modify("department_dept_name.Update = No")

   dw_1.Modify("department_dept_id.Update = No")

   dw_1.Modify("department_dept_id.Key = No")

   // Rendre possible la mise à jour de la table employée.

   dw_1.Modify( &

      "DataWindow.Table.UpdateTable = ~"employee~"")

   //Mettre à yes les colonnes pour lesquelles on désire faire la mise à jour

   dw_1.Modify("employee_emp_id.Update = Yes")

   dw_1.Modify("employee_emp_fname.Update = Yes")

   dw_1.Modify("employee_emp_lname.Update = Yes")

   dw_1.Modify("employee_emp_id.Key = Yes")

   //Faire la mise à jour de la table employé

   rc = dw_1.Update()

   IF rc = 1 THEN

      COMMIT USING SQLCA;

   ELSE

      ROLLBACK USING SQLCA;

      MessageBox("Status",  &

      + "Update of employée table failed. "  &

      + "Rolling back all changes.")

   END IF

ELSE

   ROLLBACK USING SQLCA;

   MessageBox("Status",  &

      + "Update of department table failed. "  &

      + "Rolling back changes to department.")

END IF

Ajout de la clause WHERE

Le script suivant ajoute dynamiquement une clause Where a un objet DataWindow créé avec l’option SELECT statement qui n’avait pas inclus de clause WHERE.

Etant donné que cet exemple concatène une clause where à la clause SELECT originale, des instructions auraient été nécessaires pour supprimer une clause where originale s’il y en avait eu une.

Cette technique est utile lorsque les arguments dans la clause where peuvent être modifiés durant le déroulement du programme.

L’instruction SELECT originale peut être:

SELECT employee.emp_id, employee.l_name


FROM employée

Si l’application construit une clause where basée sur les choix de l’utilisateur :

La clause WHERE peut être

WHERE emp_id > 40000

Le script de l’événement Open de la fenêtre stocke la clause SELECT originale dans original_select, une variable d’instance.

dw_emp.SetTransObject(SQLCA)

original_select = dw_emp.Describe("DataWindow.Table.Select")

Le script de l’événement Clicked d’un CommandButton concatène une clause where manipulée par la variable d’instance where_clause à la variable original_select et l’affecte à l’attribut Table.Select de la DataWindow.

string rc, mod_string

mod_string = &

"DataWindow.Table.Select='" + original_select + where_clause + "'"

rc = dw_emp.Modify(mod_string)

IF rc = "" THEN


dw_emp.Retrieve()

ELSE


MessageBox("Status", "Modify Failed" + rc)

END IF

Insertion des marques de Quotes dans la DataWindow painter

Pour les bases Watcom, ORACLE, et ALLBASE, le DataWindow painter met des guillemets autour des noms de tables et des noms de colonne.

SELECT "EMPLOYEE"."EMP_LNAME"

Sauf si les marques de quotes ont été enlevées, la clause where doit aussi utiliser les guillemets autour des noms de tables et des noms de colonnes.

where_clause = " where ~~~"EMPLOYEE~~~".~~~"SALARY~~~" > 40000"

Note : Les noms de propriétaire ne se mettent pas entre guillemets.

Mode de Requête et Spécification des Critères.

Le mode de requête permet une visualisation de la DataWindow qui permettra à l’utilisateur de spécifier les conditions de sélection de données(QBE query by Example) .

PowerBuilder construit la clause WHERE en fonction des spécifications.

Lorsque l’utilisateur quitte le mode de requête(Commute en mode visualisation des données), on reconstitue les données(Retrieve data) basées sur la clause SELECT modifiée.

Dans l’exemple qui suit une fenêtre qui affiche un contrôle DataWindow a un menu qui inclut une commande de menu appelée Select Data.

Quand l’utilisateur choisit cette commande, son script affiche la DataWindow en mode requête et met le statut de la commande du menu en mode checked.

Lorsque l’utilisateur choisit à nouveau la commande du menu, le script de la commande du menu met off le mode requête et reconstitue les données de la DataWindow basées sur la nouvelle clause WHERE spécifiée par l’utilisateur à travers le mode de requête.

Le script est aussi associé a un CheckBox libellé Sort data visible, qui permet la commutation du mode de tri on ou off.

Le script de la commande de menu Select Data est :

string rtn

IF m_selectdata.Checked = FALSE THEN

   // Mode de requête mis à ON permettant de spécifier ces critères.

   rtn = dw_1.Modify("DataWindow.QueryMode=YES")

   IF rtn = "" THEN

      // Si La fonction s’est bien déroulée, cocher sur le MenuItem

      // Le mode de requête est ON et affichage de la CheckBox de Tri

      This.Check()

      ParentWindow.cbx_sort.Show()

   ELSE

      MessageBox("Error", "Can't access query mode to select data.")

   END IF

ELSE

   // Mettre Off le mode de Requête et Reconstituer les données

   // basées sur les choix de l’utilisateur

   rtn = dw_1.Modify("DataWindow.QueryMode=NO")

   IF rtn = "" THEN

      // Si La fonction s’est bien déroulée, décocher MenuItem

      // le mode de requête est OFF , Cacher le CheckBox de Tri

      // et reconstituer les données

      This.UnCheck()

      ParentWindow.cbx_sort.Hide()

      dw_1.Retrieve()

   ELSE

      MessageBox("Error", "Failure exiting query mode.")

   END IF

END IF

Une version simple du script de l’événement clicked du CheckBox de tri suit. On peut ajouter des tests permettant de savoir si la fonction s’est déroulé avec succès ou pas.

IF This.Checked = TRUE THEN

   dw_1.Modify("DataWindow.QuerySort=YES")

ELSE

   dw_1.Modify("DataWindow.QuerySort=NO")

END IF

Spécification des critères en mode requête

"Prompt for criteria" est un autre moyen permettant à l’utilisateur de spécifier des critères de recherche.

Lorsqu’un script va rechercher des données, PowerBuilder affiche la fenêtre de spécification de critères de recherche qui offre la possibilité de spécifier des critères pour tous les colonnes qui ont été positionnées.

Les définitions suivantes vont inclure les colonnes emp_name, emp_salary, and dept_id dans la boite de Dialogue de spécification de critères lorsque la fonction Retrieve est appelée :

dw_1.Modify("emp_name.Criteria.Dialog=YES")

dw_1.Modify("emp_salary.Criteria.Dialog=YES")

dw_1.Modify("dept_id.Criteria.Dialog=YES")

Retrieve as needed

Dans l’exemple suivant, l’objet DataWindow object a été défini avec le mode Retrieve Only As Needed.

Lorsque cette option est active PowerBuilder recherche le nombre maximum d’enregistrements pour remplir la DataWindow, les affiche rapidement, et attend que l’utilisateur essaye d’afficher d’autres enregistrement pour rechercher d’autres enregistrements.

Si l’on veut le mode d’affichage rapide initial, et que l’on ne veut pas par la suite laisser des curseurs ouverts sur le serveur, on peut désactiver le mode Retrieve Only As Needed avec la fonction modify.

Après que plusieurs enregistrement aient rempli la DataWindows, le code suivant de l’événement RetrieveRow change l’attribut Retrieve.AsNeeded qui force le reste des enregistrement a être rapporté :

dw_1.Modify("DataWindow.Retrieve.AsNeeded=NO")

Changer la source de données

L’exemple suivant change la source de données de l’objet DataWindow d’une instruction SQL à une procédure stockée.

Cette technique fonctionne si seulement l’ensemble résultat n’est pas modifié(c’est à dire le nombre, type, et ordre des colonnes doivent être les mêmes pour les deux sources).

Lorsque l’on défini l’objet DataWindow, on doit prédéfinir tous les arguments de recouvrement(retrieval) de la DataWindow.

Dans cet exemple, la clause SELECT definie dans le painter a trois arguments:

une de type string, une de type number, et une autre de type date.

La procédure stockée a deux arguments, de type string,.

Dans le painter, on doit définir quatre arguments dans la DataWindow, deux de type string, une de type nombre, et une de type date.(On n’est pas obligé d’utiliser tous les arguments que l’on a définis) :

string rc, mod_string, name_str = "Watson"

integer dept_num = 100

// Enlever L’instruction SELECT de la DataWindow

Dw_1.Modify("DataWindow.Table.Select = ''")

// Définir l’attribut de procédure à votre procédure

mod_string = "DataWindow.Table.Procedure = &

   '1 exécute dbo.emp_arg2;1 @dept_id_arg &

   = :num_arg1, @lname_arg = :str_arg1'"

rc = dw_1.Modify(mod_string)

// Si les modifications sont acceptées. Rechercher les données.

IF rc = "" THEN

   dw_1.Retrieve(dept_num, name_str)

ELSE

   MessageBox("Status",  &

   "Change to DW Source Failed " + rc)

END IF

Ajout et Suppression de donnée dans l’objet DataWindow

Cette instruction supprime un objet Bitmap logo de la DataWindow dw_cust:

dw_cust.Modify("DESTROY logo")

Cette instruction supprime la colonne nommée salary de la DataWindow dw_cust.

Noter l’utilisation du mot clé Column qui permet de supprimer la colonne dans la DataWindow, et le buffer de données associé à la colonne.

dw_cust.Modify("DESTROY Column salary")

Cet exemple ajoute un rectangle nommé rect1 dans l’espace header de la DataWindow dw_cust.

string modstring

modstring = &

   'CREATE rectangle(Band = background X="206" Y="6" height="69" &

   width="1363" brush.hatch="6" brush.color="12632256" pen.style="0" &

   pen.width="14" pen.color="268435584" background.mode="2" &

   background.color="-1879048064" name=rect1) '

dw_cust.Modify(modstring)

Cette instruction ajoute un Bitmap nommé logo dans l’espace header du niveau de groupement 1 de la DataWindow dw_cust.

string modstring

modstring = &

   'CREATE Bitmap(band=footer x="37" y="12" height="101" &

   width="1509" filename="C:\stage\BEACH.BMP" border="0" name=bmp1) '

dw_cust.Modify(modstring)

Syntaxe pour créer l’objet

Pour créer un objet, on doit utiliser la syntaxe de la DataWindow comme spécifié ci dessus.

 La manière la plus facile pour avoir la bonne syntaxe pour tous les attributs nécessaires est de dessiner l’objet en se servant de la DataWindow painter et exporter la syntaxe dans un fichier dans lequel on fera toutes les modifications désirées.

Et on inclura par la suite le code dans un script comme indiqué ci dessus.

Cette manière est la plus adéquate pour obtenir des syntaxes correctes pour des objets complexes comme des graphs.

